GTL Template language

Jean-Luc Béchennec

August 19, 2016

Contents

1	Abo	out this	s document	3
2	Dat	a types	5	4
	2.1		tors priority	5
	2.2		l operators	5
		2.2.1	The exists operator	5
		2.2.2	The exists default () operator	5
		2.2.3	The typeof operator	6
		2.2.4	The mapof operator	6
		2.2.5	The listof operator	8
	2.3	Getters	s applicable to any data type	8
		2.3.1	The type getter	8
		2.3.2	The isANumber getter	8
		2.3.3	The description getter	9
	2.4	Setters	s applicable to any data type	9
		2.4.1	The setDescription setter	9
		2.4.2	The touch setter	9
	2.5	The ir	nt data type	9
		2.5.1	The int operators	9
		2.5.2	The int getters	10
		2.5.3	The int setters	13
		2.5.4	The int functions	14
	2.6	The ch	nar data type	15
		2.6.1	The char operators	15
		2.6.2	The char getters	16
	2.7	The er	num data type	17
		2.7.1	The enum operators	17
		2.7.2	The enum getters	17
	2.8	The fl	loat data type	17
		2.8.1	The float operators	17
		2.8.2	The float getters	18
		2.8.3	The float function	19
	2.9	The st	cring data type	19
		2.9.1	The string operators	19
		2.9.2	The string getters	20
		2.9.3	The string setter	24
		2.9.4	The string functions	25

	2.10	The bool data type
		2.10.1 The bool operators
		2.10.2 The bool getters
	2.11	The struct data type
		2.11.1 The struct operators
		2.11.2 The struct getter
	2.12	The list data type
		2.12.1 The list operators
		2.12.2 The list getters
	2.13	The map data type
		2.13.1 The map operators
		2.13.2 The map getters
	2.14	The set data type
		2.14.1 The set operators
		2.14.2 The set getters
		2.14.3 The set setters
	2.15	The type data type
3		L instructions 37
	3.1	The %% instruction
	3.2	The let instruction
	3.3	The unlet instruction
	3.4	The ! instruction
	3.5	The ? instruction
	3.6	The tab instruction
	3.7	The sort instruction
	3.8	The if instruction
	3.9	The foreach instruction
		The for instruction
		The loop instruction
		The repeat instruction
		The write instruction
	3.14	The template instruction
		3.14.1 Template invocation with a copy of variables
		3.14.2 Template invocation with arguments
	0.15	3.14.3 Conditional invocation of templates
		The input instruction
		The error and warning instructions
		The print and println instructions
		The display instruction
	3.19	The variables instruction
4	GTI	L modules 54
	4.1	Importing a module
	4.2	Writing a module
		4.2.1 The arguments
		4.2.2 Function definition
	4.3	Getter definition
	4.4	Setter definition

Chapter 1

About this document

This document presents the syntax and the semantics of GTL 3, a template language used in Goil, the OIL compiler of Trampoline¹, to generate code from the OIL or arXML description of an OSEK/AUTOSAR application.

GTL is written in GalGas 3², a lexical and syntactic analyzer, which is also a powerful domain specific language to write compilers and interpreters. GalGas 3 is developed by Pierre Molinaro from the Real-Time Systems Group of IRCCyN, a french laboratory from a joint of the University of Nantes, École Centrale de Nantes, École des Mines de Nantes and CNRS. The first version of the GTL interpreter was written by Pierre Molinaro too. As needs have increased, more features was needed and this led to a major rewrite of the GTL interpreter; In fact no original code made its way in GTL 3.

Convention

Code examples follow the following conventions:

- GTL example code snippets are presented in light blue boxes. In these examples, *Italic* writing are used for generic syntactic items. For instance, expression means any expression like a + b or "a string". Boldface words are reserved words of GTL, like foreach or built-in functions, getters or setters like setDescription. Pieces of code delimited by < and > are optional.
- Standard output of examples are presented in light yellow boxes.
- Template string output of examples are presented in light green boxes.

¹An OSEK and AUTOSAR 4.2 compliant RTOS, check https://github.com/TrampolineRTOS/ ${}^2 Gal Gas \ 3 \ is \ free \ software \ distributed \ under \ the \ GPL \ license \ and \ can \ be \ found \ at \ http://galgas.$

rts-software.org

Chapter 2

Data types

GTL supports the following data types:

\mathbf{type}	Description		
int	arbitrary precision integer numbers. The GMP library is used		
char	unicode chars		
float	64 bits floating point numbers		
bool	standard boolean		
enum	enumerated type		
string	unicode strings		
struct	structured data		
list	lists of data, may be accessed as a table		
map	map (aka dictionary) of data		
set	set of strings		
type	the type of a data		
unconstructed	an unconstructed variable		

Data embed a location, that is a file name, a line and a column. Each time a variable is set, the corresponding location is set too. This allow to report errors efficiently. See 2.4.2 and 3.16.

Data embed a description string too. This allow to comment on the content of the data. See 2.3.3 and 2.4.1.

Each type has its set of operators, getters and setters. The expression, for getters, or the variable, for setters, is called the *target*. Getters return a value related to the expression but do not change it. They are used to get information from a data or to convert it into another type. Setters may only target a variable. Setters change the content of the variable and do not return anything. Getters and setters may have arguments. Syntax for getters without argument is as follow:

```
[expression aGetter]
```

When the getter takes arguments, they are listed after a colon and separated commas as follow:

```
[expression aGetter: arg1, arg2, ..., argN]
```

Syntax for setters without arguments is as follow:

```
[!variable aSetter]
```

When the setter takes arguments, they are listed after a colon and separated by commas as follow:

```
[!variable aSetter: arg1, arg2, ..., argN]
```

2.1 Operators priority

The following table gives all the operators available in GTL an their priority. Semantics of the operators is given for each data type. Operators of same priority are evaluated from left to right.

Priority	Operators
0	^
1	&
2	== != < > <= >=
3	<< >> +
4	* / mod
5	not ~ - +

2.2 Special operators

2.2.1 The exists operator

The **exists** operator tests the existence of a variable, a struct field, an item of a map or a list and returns a bool, **true** if the entity exists, **false** otherwise.

```
exists var
```

Example

```
let c := 3
if exists c then println "c exists" else println "c does not exist" end if
unlet c # delete c
if exists c then println "c exists" else println "c does not exist" end if

c exists
c does not exist
```

2.2.2 The exists ... default (...) operator

The exists ... default (...) operator tests the existence of a variable, a struct field, an item of a map or a list.

```
exists var default ( expression )
```

If the entity exists it is returned. Otherwise the evaluation of the default expression is returned.

```
let aList := @( 1, 2, 3 )
let aSecondList := exists aList default ( @() )
display aSecondList
unlet aList # delete aList
let aSecondList := exists aList default ( @() )
display aSecondList
```

```
aSecondList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 9:7
  - list: @(
   0 :>
       integer: 1
       integer: 2
   2 :>
       integer: 3
aSecondList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 12:7
 - list: @(
```

The typeof operator

The typeof operator takes a variable as argument an return its type.

```
typeof var
```



The typeof is deprecated. Use the type getter instead. See 2.3.1.

The mapof operator 2.2.4

The mapof may apply on struct or on list.

With a struct

The mapof operator converts a struct to a map.

```
mapof expression end
```



mapof is deprecated, use the map getter on struct instead. See 2.11.2.

```
let aStruct := @{ a: 1, b: 2, c: 3 }
display aStruct
let aMap := mapof aStruct end
display aMap
```

```
aStruct from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 3:7
 - struct: @{
   a :>
     integer: 1
```

```
b :>
    integer: 2
c :>
    integer: 3
}
aMap from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 5:7
- map: @[
    "a" :>
    integer: 1
    "b" :>
    integer: 2
    "c" :>
    integer: 3
]
```

With a list

The **mapof** operator converts a list to a map according to a string field of each item of the list. So the list should be a list of struct.

```
mapof var by identifier
```



mapof ... by is deprecated, use the mapBy getter on list instead. See 2.12.2.

```
aMap from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 13:7
  - map: @[
   "Arnold" :>
       struct: @{
           age :>
               integer: 18
           height :>
               integer: 180
           name :>
              string: "Arnold"
   "Bob" :>
       struct: @{
           age :>
               integer: 22
           height :>
              integer: 170
           name :>
               string: "Bob"
       }
    "John" :>
       struct: @{
          age :>
```

```
integer: 29
height :>
    integer: 175
name :>
    string: "John"
}
```

2.2.5 The listof operator

The listof operator apply to a map variable. It returns a list representation of the map. The elements of the list are sorted in the alphanumerical order of the map keys.

```
listof var end
```



listof ... by is deprecated, use the list getter on map instead. See 2.13.2.

2.3 Getters applicable to any data type

2.3.1 The type getter

The type getter returns the type of the expression. See the section 2.15.

Example

2.3.2 The isANumber getter

isanumber returns true if the expression is a number: int or float, false otherwise. This getter is useful to test the type of an argument in a function, a getter or a setter.

```
let b := 0
let a := 1
if [a isANumber] then
  let b := a # if a is a number, it is copied in b
else
  let b := 0 # otherwise b is set to 0
end if
display b
```

```
b from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 13:7
- integer: 1
```

2.3.3 The description getter

description returns the string describing the data, if available, or an empty string otherwise. If the data is coming from an OIL source file, it corresponds to the description field (see section 2.3.11 of System Generation – OIL: OSEK Implementation Language – Version 2.5).

2.4 Setters applicable to any data type

2.4.1 The setDescription setter

setDescription takes one string argument desc. It sets the string describing the data to desc.

2.4.2 The touch setter

touch sets the location of modification of the data to the current one.

2.5 The int data type

The int data type support arbitrary precision arithmetic by using the GNU Multiple Precision Arithmetic Library (GMP).

2.5.1 The int operators

The int datatype supports the following operators:

Unary operators

Operator	Expression type	Meaning
+	int ← +int	Plus operator. No effect
_	int ← -int	Minus operator. Negation
~	int ← ~int	Not operator. Complementation by 1

Binary arithmetic operators

Operator	Expression type	Meaning	
+	$int \leftarrow int + int$	Addition	
_	$int \leftarrow int - int$	Substraction	
*	$int \leftarrow int * int$	Multiplication	
/	$int \leftarrow int / int$	Division	
mod	$int \leftarrow int mod int$	Modulus	

Binary bitwise operators

Operator	Expression type	Meaning	
&	$int \leftarrow int \& int$	bitwise and	
	$int \leftarrow int \mid int$	bitwise or	
^	$int \leftarrow int ^ int$	bitwise exclusive or	

Operator	Expression type	Meaning
<<	$int \leftarrow int << int$	shift left
>>	$int \leftarrow int >> int$	shift right

Comparison operators

Operator	Meaning	
!=	bool ← int != int	Not equal
==	bool ← int == int	Equal
>	$bool \leftarrow int > int$	Greater than
<	$bool \leftarrow int < int$	Lower than
>=	bool ← int >= int	Greater or equal
<=	bool ← int <= int	Lower or equal

2.5.2 The int getters

The string getter

string returns a string decimal representation of the int expression.

Example

The hexString getter

hexstring returns a hexadecimal string representation of the int expression prefixed by "0x". If the int expression is negative a '-' is inserted before.

Example

```
let a := [42 hexString]
display a
let b := [-20 hexString]
display b

a from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 21:7
    - string: "0x2A"
b from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 23:7
    - string: "-0x14"
```

the xString getter

xstring returns a hexadecimal string representation of the int expression. If the expression is negative a '-' is inserted before.

```
let a := [42 xString]
display a
let b := [-20 xString]
display b

a from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 26:7
    - string: "2A"
b from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 28:7
    - string: "-14"
```

The numberOfBytes getter

numberOfBytes returns an int, the number of bytes needed to store an unsigned expression.

Example

```
println [255 numberOfBytes]
println [256 numberOfBytes]

1
2
```

The signedNumberOfBytes getter

signedNumberOfBytes returns an int, the number of bytes needed to store a signed expression.

Example

```
println [127 signedNumberOfBytes]
println [128 signedNumberOfBytes]
```

The numberOfBits getter

numberOfBits returns an int, the number of bits needed to store an unsigned expression.

```
println [63 numberOfBits]
println [64 numberOfBits]
6
7
```

The signedNumberOfBits getter

signedNumberOfBits returns an int, the number of bits needed to store a signed expression.

Example

```
println [63 signedNumberOfBits]
println [64 signedNumberOfBits]

7
8
```

The sign getter

sign returns an int, -1 if the expression is strictly negative, 0 if it is null and +1 if the expression is strictly positive.

The fitsUnsignedInByte getter

fitsUnsignedInByte returns a bool, true if the expression fits in an unsigned byte, false otherwise.

The fitsSignedInByte getter

fitsSignedInByte returns a bool, true if the expression fits in a signed byte, false otherwise.

The fitsUnsignedInWord getter

fitsUnsignedInWord returns a bool, true if the expression fits in an unsigned 16 bits word, false otherwise.

The fitsSignedInWord getter

fitsSignedInWord returns a bool, true if the expression fits in a signed 16 bits word, false otherwise.

The fitsUnsignedInLong getter

fitsUnsignedInLong returns a bool, true if the expression fits in an unsigned 32 bits long, false otherwise.

The fitsSignedInLong getter

fitsSignedInLong returns a bool, true if the expression fits in a signed 32 bits long, false otherwise.

The fitsUnsignedInLongLong getter

fitsUnsignedInLongLong returns a bool, true if the expression fits in an unsigned 64 bits long long, false otherwise.

The fitsSignedInLongLong getter

fitsSignedInLongLong returns a bool, true if the expression fits in a signed 64 bits long long, false otherwise.

The abs getter

abs returns an int, the absolute value of the expression.

The bitAtIndex getter

bitAtIndex takes one int argument: index. It returns true if the bit at index is set and false otherwise. index 0 corresponds to the lowest significant bit.

Example

2.5.3 The int setters

The setBitAtIndex setter

setBitAtIndex takes two arguments. The first one, value, is a bool. The second one, index, is the index of the bit to set. if value is true the bit is set to 1 and to 0 otherwise.

Example

The complementBitAtIndex setter

complementBitAtIndex takes one int argument, index, which is the index of the bit to complement.

2.5.4 The int functions

The following built-in functions return an int.

The function majorVersion()

majorVersion() takes no argument and returns the major version of the compiler in where GTL is embedded. For instance, if the version string is "3.2.1", it returns 3.

The function minorVersion()

minorVersion() takes no argument and returns the minor version of the compiler in where GTL is embedded. For instance, if the version string is "3.2.1", it returns 2.

The function revision()

revision() takes no argument and returns the revision of the compiler in where GTL is embedded. For instance, if the version string is "3.2.1", it returns 1.

The function max8bitsUnsignedInt()

max8bitsUnsignedInt() takes no argument and returns the maximum unsigned number that fits in 8 bits.

The function max8bitsSignedInt()

max8bitsSignedInt() takes no argument and returns the maximum signed number that fits in 8 bits.

The function min8bitsSignedInt()

min8bitsSignedInt() takes no argument and returns the minimum signed number that fits in 8 bits.

The function max16bitsUnsignedInt()

max16bitsUnsignedInt() takes no argument and returns the maximum unsigned number that fits in 16 bits.

The function max16bitsSignedInt()

max16bitsSignedInt() takes no argument and returns the maximum signed number that fits in 16 bits.

The function min16bitsSignedInt()

min16bitsSignedInt() takes no argument and returns the minimum signed number that fits in 16 bits.

The function max32bitsUnsignedInt()

max32bitsUnsignedInt() takes no argument and returns the maximum unsigned number that fits in 32 bits.

The function max32bitsSignedInt()

max32bitsSignedInt() takes no argument and returns the maximum signed number that fits in 32 bits.

The function min32bitsSignedInt()

min32bitsSignedInt() takes no argument and returns the minimum signed number that fits in 32 bits.

The function max64bitsUnsignedInt()

max64bitsUnsignedInt() takes no argument and returns the maximum unsigned number that fits in 64 bits.

The function max64bitsSignedInt()

max64bitsSignedInt() takes no argument and returns the maximum signed number that fits in 64 bits.

The function min64bitsSignedInt()

min64bitsSignedInt() takes no argument and returns the minimum signed number that fits in 64 bits.

2.6 The char data type

The char data type supports unicode characters. A literal char is delimited by a pair of ':

Literal chars support escaped special characters:

Escape sequence	Corresponding character
\f	form feed
\n	new line
\r	return
\t	horizontal tab
\v	vertical tab
\\	backslash
\'	,
\Ø	null character
\u <i>nnnn</i>	unicode character with code <i>nnnn</i> in hexadecimal
$\backslash Unnnnnnn$	unicode character with code nnnnnnn in hexadecimal

2.6.1 The char operators

The char data type supports the following operators:

Comparison operators

Operator	Meaning	
! =	bool ← char != char	Not equal
==	$bool \leftarrow char == char$	Equal
>	bool ← char > char	Greater than
<	$bool \leftarrow char < char$	Lower than
>=	bool ← char >= char	Greater or equal
<=	bool ← char <= char	Lower or equal

2.6.2 The char getters

The string getter

string returns a string representation of the char expression. ['c'] string returns string "c".

The isAlnum getter

isAlnum returns a bool, true if the char expression is an ASCII alphanumeric character: between 'A' and 'Z' or between 'A' and 'Z' or between 'A' and 'Z' or between 'A' and 'B' and '

The isAlpha getter

isAlpha returns a bool, true if the char expression is an ASCII letter: between 'A' and 'Z' or between 'a' and 'Z', false otherwise.

The isDigit getter

isDigit returns abool, true if the char expression is an ASCII digit: between '0' and '9', false otherwise.

The isCntrl getter

isCntrl returns a bool, true if the char expression is an ASCII control character, i.e. strictly before the SPACE character, false otherwise.

The isLower getter

isLower returns a bool, true if the char expression is an ASCII lower case letter: between 'a' and 'z', false otherwise.

The isUpper getter

isUpper returns a bool, true if the char expression is an ASCII upper case letter: between 'A' and 'Z', false otherwise.

The isXDigit getter

isxDigit returns a bool, true if the char expression is an ASCII hexadecimal digit: between 'A' and 'F' or between 'A' and 'f' or between 'O' and '9', false otherwise.

2.7 The enum data type

The enum data type allow to store identifiers in variables. A literal enum begins with a '\$' followed with a name composed of ASCII letters, numbers and '_'.

Example

```
let a := $auto
display a

a from file '/Users/jlb/Develop/GTL/examples/enumTest.gtl', line 4:7
    enum: auto
```

2.7.1 The enum operators

Comparison operators

Operator	Expression type	Meaning
!=	$\texttt{bool} \leftarrow \texttt{enum} \ !\texttt{=} \ \texttt{enum}$	Not equal
==	$\texttt{bool} \leftarrow \texttt{enum} == \texttt{enum}$	Equal

2.7.2 The enum getters

The string getter

string returns a string representation of the enum expression. [@auto string] returns string
"auto".

2.8 The float data type

The float data type is the standard IEEE784 64 bits floating point number.

2.8.1 The float operators

The float data type supports the following operators:

Unary operators

Operator	Expression type	Meaning
+	float ← +float	Plus operator. No effect
_	float ← -float	Minus operator. Negation

Binary arithmetic operators

Operator	Expression type	Meaning
+	$float \leftarrow float + float$	Addition
_	$float \leftarrow float - float$	Substraction
*	float ← float * float	Multiplication

Operator	Expression type	Meaning
/	float ← float / float	Division

Comparison operators

Operator	Expression type	Meaning
! =	bool ← float != float	Not equal
==	$bool \leftarrow float == float$	Equal
>	bool ← float > float	Greater than
<	$bool \leftarrow float < float$	Lower than
>=	bool ← float >= float	Greater or equal
<=	bool ← float <= float	Lower or equal

2.8.2 The float getters

The string getter

string returns a string representation of the float expression. [4.2 string] returns string
"4.2".

The cos getter

cos returns the cosine of a float expression expressed in radian.

The sin getter

sin returns the sine of a float expression expressed in radian.

The tan getter

tan returns the tangent of a float expression expressed in radian.

The cosDegree getter

cosDegree returns the cosine of a float expression expressed in degree.

The sinDegree getter

sinDegree returns the sine of a float expression expressed in degree.

The tanDegree getter

tanDegree returns the tangent of a float expression expressed in degree.

The exp getter

exp returns the exponentiation of a float expression.

The logn getter

logn returns the natural logarithm of a float expression.

The log2 getter

10g2 returns the logarithm base 2 of a float expression.

The log10 getter

log10 returns the logarithm base 10 of a float expression.

The sqrt getter

sqrt returns the square root of a float expression.

The power getter

power takes one float argument, p. It returns the expression raised to the power of p.

2.8.3 The float function

The following function returns a float

The pi() function

pi() returns an approximation of the π constant value (3.141592653589793238462643383279 50288).

2.9 The string data type

The string data type supports unicode. A literal string is delimited by a pair of ".

```
let a := "A literal string"
```

Literal strings support escaped special characters:

Corresponding character
form feed
new line
return
horizontal tab
vertical tab
backslash
,
"
null character
unicode character with code <i>nnnn</i> in hexadecimal
unicode character with code nnnnnnn in hexadecimal

2.9.1 The string operators

The string data type supports the following operators:

Binary operator

Operator	Expression type	Meaning
+	$string \leftarrow string + string$	Concatenation

Comparison operators

Operator	Expression type	Meaning
! =	$bool \leftarrow string != string$	Not equal
==	$bool \leftarrow string == string$	Equal
>	$bool \leftarrow string > string$	Greater than
<	$bool \leftarrow string < string$	Lower than
>=	$bool \leftarrow string >= string$	Greater or equal
<=	$bool \leftarrow string <= string$	Lower or equal

2.9.2 The string getters

The charAtIndex getter

charAtIndex takes one int argument, the index. It returns a char, the character at index index. The first char is at index 0, the last char is at index length of the string minus 1. If index is greater or equal than the length of the target, a run time error occurs.

Example

The $indexOfChar\ getter$

indexOfChar takes one char argument, the character to look up. It returns an int, the index of the first occurrence of character in the target. If character is not found in the target, -1 is returned.

The indexOfCharInRange getter

indexOfCharInRange takes two char argument, minChar and maxChar which define the character range to look up. It returns an int, the index of the first occurrence of the character being in range (bounds included) in the target. If character is not found in the target, -1 is returned.

Example

The containsChar getter

containsChar takes one char argument, the character to look up. It returns a bool, true is the target contains the character, false otherwise.

The containsCharInRange getter

containsCharInRange takes two char argument, minchar and maxchar which define the character range to look up. It returns a bool, true is the target contains a character within the range (bounds included), false otherwise.

The HTMLRepresentation getter

HTMLRepresentation returns a representation of the string suitable for an HTML encoded representation. '&' is encoded by & , '"' by " , '<' by < and '>' by >.

The identifierRepresentation getter

identifierRepresentation returns an unique representation of the string conforming to a C identifier. Any Unicode character that is not a latin letter is transformed into its hexadecimal code point value, enclosed by '_' characters. This representation is unique: two different strings are transformed into different C identifiers. For example: value3 is transformed to value_33_; += is transformed to _2B__3D_; An_Identifier is transformed to An_5F_-Identifier.

The fileExists getter

fileExists returns a bool, true if a file exists at the target path, false otherwise.

The length getter

length returns and int, the number of characters in the string.

The lowercaseString getter

lowercaseString returns the lowercased representation of the string.

The capitalized getter

If the string is empty, **capitalized** returns the empty string; otherwise, it returns the string with the first character being replaced with the corresponding uppercase character.

The uppercaseString getter

uppercaseString returns the uppercased representation of the target.

The leftSubString getter

leftSubString takes one int argument, number, and returns the sub-string from the beginning of the target and with, number, of characters. If the sub-string is longer than the target, the target is returned.

Example

```
let str := ["Hello World !" leftSubString : 5]
display str

str from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 85:7
    - string: "Hello"
```

The rightSubString getter

rightsubstring takes one int argument, number, and returns the sub-string from the end of the target and with number of characters. If the sub-string is longer that the target, the target is returned.

Example

```
let str := [["Hello World !" leftSubString : 11] rightSubString: 5]
display str

str from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 88:7
    - string: "World"
```

The subString getter

substring returns the sub-string from the index passed as first argument and with the number of characters passed as second argument. If the index is out of the target, the empty string is returned. If the number of characters is greater than the available sub-string, the sub-string is returned.

```
let firstStr := ["Hello World !" subString : 6, 5]
display firstStr
let secondStr := ["Hello" subString : 10, 3]
display secondStr
let thirdStr := ["Hello" subString : 2, 10]
display thirdStr
```

```
firstStr from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 91:7
  - string: "World"
secondStr from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 93:7
  - string: ""
thirdStr from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 95:7
  - string: "llo"
```

The reversedString getter

reversedString returns a mirrored string.

Example

```
let str := ["Hello World !" reversedString]
display str

str from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 98:7
    - string: "! dlroW olleH"
```

The componentsSeparatedByString getter

componentsSeparatedByString takes one string argument: separator. The target is cut into string pieces according to the separator and a list of the pieces is returned.

Example

```
let componentList := ["Hello World !" componentsSeparatedByString : " "]
display componentList

componentList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 101:7
   - list: @(
   0 :>
        string: "Hello"
   1 :>
        string: "World"
   2 :>
        string: "!"
)
```

The ${\tt columnPrefixedBy}\ {\tt getter}$

columnPrefixedBy takes one string argument, prefix, and returns the target with each line prefixed by prefix.

```
let formattedStr := ["Hello\nWorld" columnPrefixedBy : "# "]
println formattedStr

# Hello
# World
```

The wrap getter

wrap wraps the target to a width. This getter takes two int arguments: width and shift. The target is assumed to contain paragraphs separated by '\n'. It returns the target with each paragraph wrapped to width. In addition, each line of the paragraph except the first one is prefixed by shift spaces.

Example

```
let wrappedStr := ["Hello beautiful World.\nHow are you" wrap : 6, 2]
println wrappedStr

Hello
   beautiful
   World.
How
   are
   you
```

The subStringExists getter

subStringExists takes one argument, subString. It returns a bool, true if the sub-string subString is found in the target, false otherwise.

The replaceString getter

replaceString takes two argument, find and replace. It returns the target where each occurrence of find is replaced by replace.

The envVar getter

envVar returns a string, the value of the target environment variable. If it does not exists, envVar returns the empty string.

The envVarExists getter

envVarExists returns a bool, true if target environment variable exists, false otherwise.

2.9.3 The string setter

The setCharAtIndex setter

setCharAtIndex takes two arguments. The first one is the character to set and the second one is the index. It sets the character at index index. If <a href="https://index.character.com/c

2.9.4 The string functions

The following built-in functions return a string:

The version() function

version() takes no argument and returns the version string of the compiler in where GTL is embedded. For instance, in goil version 3, it returns "3.0.0".

The currentDir() function

currentDir() takes no argument and returns the current directory.

The homeDir() function

homeDir() takes no argument and returns the home directory.

The currentDateTime() function

currentDateTime() takes no argument and returns the current date and time in the following
format: "<dayname> <month> <daynum> <time> <year>".

Example

```
println currentDateTime()

Wed Aug 17 15:16:20 2016
```

The trueFalse() function

trueFalse() is deprecated and is replaced by the trueOrFalse getter (see 2.10.2) of the bool data type. It takes one bool argument and returns "true" or "false" according to the argument.

The TrueFalse() function

TrueFalse() takes one bool argument and returns "True" or "False" according to the argument.

The yesNo() function

yesno() is deprecated and is replaced by the yesorno getter (see 2.10.2) of the bool data type. This function takes one bool argument and returns "yes" or "no" according to the argument.

The TRUEFALSE () function

TRUEFALSE() is deprecated and is replaced by the TRUEOrFALSE getter (see 2.10.2) of the bool data type. This function takes one bool argument and returns "TRUE" or "FALSE" according to the argument.

2.10 The bool data type

A true literal bool can be written as **true** or **yes** and a false literal bool can be written as **false** or **no**.

2.10.1 The bool operators

The bool data type supports the following operators:

Unary operator

Operator	Expression type	Meaning
~	bool ← bool	logical not
not	bool ← bool	logical not

Binary operator

Operator	Expression type	Meaning
&	bool ← bool & bool	logical and
	bool ← bool bool	logical or
^	bool ← bool ^ bool	logical exclusive or

Comparison operators

For comparison operators, false is considered to be lower than true.

Operator	Expression type	Meaning
! =	bool ← bool != bool	Not equal
==	bool ← bool == bool	Equal
>	bool ← bool > bool	Greater than
<	bool ← bool < bool	Lower than
>=	bool ← bool >= bool	Greater or equal
<=	bool ← bool <= bool	Lower or equal

2.10.2 The bool getters

The trueOrFalse getter

trueOrFalse returns a string representation, "true" or "false", of the bool expression.

The string getter

string returns a string representation, "true" or "false", of the bool expression.

The yesOrNo getter

yesOrNo returns a string representation, "yes" or "no", of the bool expression.

The TRUEOFFALSE getter

TRUEOFFALSE returns a string representation, "TRUE" or "FALSE", of the bool expression.

The YESOrNO getter

YESOTNO returns a string representation, "YES" or "NO", of the bool expression.

The int getter

int returns an int representation, 1 for true or 0 for false, of the bool expression.

2.11 The struct data type

The struct data type allows to store a heterogeneous set of data in one variable. Struct members are accessed by using the :: separator. If a is a struct, a::b refers to field b of a.

A literal struct is defined as follow:

```
@{ a: 1, b: 2, c: 3 }
```

This define a struct with fields a, b and c and respective values 1, 2 and 3.

2.11.1 The struct operators

The struct data type supports the following operators:

Operator	Expression type	Meaning
! =	bool ← struct != struct	Not equal
==	bool ← struct == struct	Equal

Two structs are equal if:

- they have the same number of field
- they have the same field names
- they have the same field values

2.11.2 The struct getter

The map getter

map returns a map representation of the target.

```
let a := @{ a: 1, b: 2, c: 3 }
let b := [a map]
display a
display b
```

```
a from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 120:7
- struct: @{
    a :>
```

```
integer: 1
b :>
    integer: 2
c :>
    integer: 3
}
b from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 121:7
- map: @[
    "a" :>
    integer: 1
    "b" :>
    integer: 2
    "c" :>
    integer: 3
]
```

2.12 The list data type

The list data type allows to store a list of data. List items are accessed by using [index] where index is the rank of the element starting at 0. If a is a list, a[0] refers to element 0 of a.

A literal list is defined as follow:

```
@( 1, 2, 3 )
```

This define a list of int with elements 1 (index 0), 2 (index 1) and 3 (index 2). An empty list can be initialized using the emptylist constant.

 Λ

The emptylist constant is deprecated. Use a literal empty list, @(), instead.

2.12.1 The list operators

The list data type supports the following operators:

Binary operators

Operator	Expression type	Meaning
+	$list \leftarrow list + any$	append any at the end of the list
	list ← list list	Concatenate lists

Comparison operators

Operator	Expression type	Meaning
!=	bool ← list != list	Not equal
==	bool ← list == list	Equal

Two lists are equal if:

- they have the same number of elements
- they have the same elements values

2.12.2 The list getters

The length getter

length returns the number of elements in the list.

The first getter

first returns the first element of the list.

The last getter

last returns the last element of the list.

The mapBy getter

mapBy takes a string argument which is the field (for an item in a list of struct) or the key (for an item in a list of a map) used as key to store the element in the resulting map. It returns a map where each element is the element of the list with the key being the corresponding field/key. If any of the item does not have a corresponding field/key, a run-time error occurs.

```
let myList := @(
    @{ age : 18, height : 180, name : "Arnold"},
    @{ age : 22, height : 170, name : "Bob"},
    @{ age : 29, height : 175, name : "John"}
)

let myMap := [myList mapBy : "name"]
display myMap
```

```
myMap - map: @[
   "Arnold" :>
       struct: @{
           age :>
               integer: 18
           height :>
               integer: 180
           name :>
               string: "Arnold"
    "Bob" :>
       struct: @{
           age :>
               integer: 22
           height :>
              integer: 170
           name :>
               string: "Bob"
       }
    "John" :>
       struct: @{
           age :>
              integer: 29
           height :>
               integer: 175
           name :>
```

```
string: "John"
}
```

The set getter

set assumes the list is a list of items convertible to string. If any of the list item is not convertible to a string, a runtime error occurs. set returns a set representation of the list.

Example

```
let aList := @( 1, 2, 4, "Hello", 4, 2, 1 )
let aSet := [aList set]
display aSet

aSet from file '/Users/jlb/Develop/GTL/examples/setTest.gtl', line 30:7
    set: @!
        1, 2, 4, Hello
    !
```

The subListTo getter

subListTo takes an int argument which is the stop index of the sublist. It returns a sublist which is a copy of target list ranging from 0 to the index included. If the index is greater than of equals the length of the target, the target is returned.

Example

```
let aList := @( 1, 2, 3, 4 )
let aList := [aList subListTo: 1]
display aList

aList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 125:7
   - list: @(
    0 :>
        integer: 1
    1 :>
        integer: 2
)
```

The subListFrom getter

subListFrom takes an int argument which is the start index of the sublist. It returns a sublist which is a copy of target list ranging from index included to the end of the list.

```
let aList := @( 1, 2, 3, 4 )
let aList := [aList subListFrom: 2]
display aList
```

```
aList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 129:7
- list: @(
    0 :>
        integer: 3
1 :>
        integer: 4
)
```

The subList getter

subList takes 2 int arguments. The first one is the start index of the sublist. The second one is the length of the sublist. It returns a sublist which is a copy of target list ranging from index included to up to length items.

Example

```
let aList := @( 1, 2, 3, 4 )
let aFirstList := [aList subList: 1, 5]
display aFirstList
let aSecondList := [aList subList: 2, 1]
display aSecondList
```

```
aFirstList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 133:7
- list: @(
    0 :>
        integer: 2
    1 :>
        integer: 3
    2 :>
        integer: 4
)
aSecondList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 135:7
- list: @(
    0 :>
        integer: 3
)
```

The list setters

The insert setter

insert takes 2 arguments. The first one is the index of the list where the data will be inserted. The second one is the data to insert. It inserts data before the item at index. If index is greater than or equals the length of the list, data is appended to the list.

```
integer: 1
1 :>
    string: "Hello"
2 :>
    integer: 2
3 :>
    integer: 3
4 :>
    integer: 4
5 :>
    string: "At the end"
)
```

2.13 The map data type

The map data type allows to store an association of key and value. map members are accessed by using <code>[key]</code> where <code>key</code> is a string. If <code>a</code> is a map, <code>a["John"]</code> refers to an element of <code>a</code> having key <code>"John"</code>.

A literal map is defined as follow:

```
@[ "age" : 29, "height" : 175, "name" : "John" ]
```

An empty map can be initialized using the emptymap constant.



The emptymap constant is deprecated. Use a literal empty map, @[], instead.

2.13.1 The map operators

The map data type supports the following operators:

	Operator	Expression type	Meaning
ĺ	!=	bool ← map != map	Not equal
	==	bool ← map == map	Equal

Two maps are equal if:

- they have the same number of items
- they have the same item keys
- they have the same item values

2.13.2 The map getters

The length getter

length returns an int, the number of elements in the map.

The list getter

list returns a list representation of the map. Elements of the list are in the alphanumerical order of the keys of the map.

Example

```
let aMap := @[ "age" : 29, "height" : 175, "name" : "John" ]
let aList := [aMap list]
display aList

aList from file '/Users/jlb/Develop/GTL/examples/testTypes.gtl', line 144:7
    list: @(
        0 :>
            integer: 29
        1 :>
            integer: 175
        2 :>
            string: "John"
    )
```

2.14 The set data type

The set data type is a set of strings. Any data that can be converted to a string may be added to a set but the actual value stored in the set is the string representation of the data. Adding a data which is already in the set has no effect.

A literal set is defined as follow.

```
@! 1, 2, 3, 4, "Hello" !
```

A literal empty set is defined as follow.

```
@!!
```

2.14.1 The set operators

The set data type supports the following operators:

Binary operators

Operator	Expression type	Meaning
+	$set \leftarrow set + any$	add the string representation of any to the set
_	$set \leftarrow set_1 - set_2$	remove from set_1 all the element of set_2
	set ← set set	does the union of sets
&	set ← set & set	does the intersection of sets

Comparison operators

Operator	Expression type	Meaning
!=	bool ← set != set	Not equal
==	bool ← set == set	Equal
>	$\texttt{bool} \leftarrow \texttt{set}_1 > \texttt{set}_2$	set ₂ strictly included in set ₁
<	$\texttt{bool} \leftarrow \texttt{set}_1 < \texttt{set}_2$	set ₁ strictly included in set ₂
>=	$\texttt{bool} \leftarrow \texttt{set}_1 >= \texttt{set}_2$	set_2 included in or equal to set_1
<=	$\texttt{bool} \leftarrow \texttt{set}_1 \mathrel{<=} \texttt{set}_2$	set_1 included in or equal to set_2

2.14.2 The set getters

The length getter

length returns an int, the number of elements in the set.

Example

```
let aSet := @! 1, 2, "yes", "no" !
let len := [aSet length]
display len
```

```
len from file '/Users/jlb/Develop/GTL/examples/setTest.gtl', line 32:7
  integer: 4
```

The list getter

list getter returns a list representation of the set. Each element of the returned list is a string. Elements of the list are sorted in the alphanumerical order.

Example

The contains getter

contains getter takes one argument, item, that must convertible to a string and returns a bool, true if item is in the set, false otherwise.

```
let aSet := @! 1, "yes", 2, "no" !
let ok := [aSet contains: "yes"]
display ok
```

```
ok from file '/Users/jlb/Develop/GTL/examples/setTest.gtl', line 40:7
  boolean: true
```

2.14.3 The set setters

The add setter

add takes one argument element and add it to the target set. element must be convertible to a string. If element is already in the set, add does nothing.

The setter is redundant with the += operator but more efficient because the set is not copied in the process.

Example

```
let a := @! 1, 2, "no" !
[!a add: "yes"]
display a

a from file '/Users/jlb/Develop/GTL/examples/setTest.gtl', line 5:7
    set: @!
        1, 2, no, yes
    !
```

The remove setter

remove takes one argument, element, and removes it from the target set. element must be convertible to a string. If element is not in the set, remove does nothing.

The setter is redundant with the += operator but more efficient because the set is not copied in the process.

Example

```
let a := @! 1, 2, "no" !
[!a remove: 2]
display a

a from file '/Users/jlb/Develop/GTL/examples/setTest.gtl', line 11:7
    set: @!
        1, no
    !
```

2.15 The type data type

The type data type store the type of an expression. It can be used to check dynamically the type of the arguments passed to a function, a getter or a setter, see chapter 4.

A set of constants corresponding to each type is define. These constants begin with a '@': @int, @char, @bool, @float, @string, @enum, @struct, @list, @map, @set @unconstructed and @type.

To get the type of any expression, use the type getter. See 2.3.1.

```
let a := 4
if [a type] == @int then
let a += 1
end if
```

```
display a
let a := [a string]
if [a type] == @string then
  let a += "1"
end if
display a
```

```
a from file '/Users/jlb/Develop/GTL/examples/typeTest.gtl', line 14:7
   integer: 5
a from file '/Users/jlb/Develop/GTL/examples/typeTest.gtl', line 19:7
   string: "51"
```

Chapter 3

GTL instructions

3.1 The %...% instruction

The \%...\% is the literal template string instruction. Every character appearing between the '%' are accumulated in the output string of the template. GTL starts by assuming a '%' exists just before the first character of the file. So if the first character of the file is not a '%' the first instruction is a \%...\% instruction up to the first '%' in the file.

Example

```
# Assuming we start in code mode
%This is the output of a template
%
foreach item in ["Hello World !" componentsSeparatedByString: " "]
do    !item
between%
%
end foreach%
%
This is the output of a template
Hello
World
!
```

3.2 The let instruction

let is the variable assignment instruction. The general form is:

```
let var := expression

If the variable does not exists, it is created. The variable is set to expression
If the := expression is omitted, the variable is created and is unconstructed:
let var
```

As in the C language, GTL has assignment operators. For instance to increment an int variable, one can write:

The following table gives the available assignment operators and their meaning.

Assign.	int	float	string	bool	struct	\mathbf{list}	map	uncons
+=	+	+	concat	NA	NA	append	NA	NA
-=	_	_	NA	NA	NA	NA	NA	NA
*=	*	*	NA	NA	NA	NA	NA	NA
/=	/	/	NA	NA	NA	NA	NA	NA
mod=	mod	NA	NA	NA	NA	NA	NA	NA
<<=	<<	NA	NA	NA	NA	NA	NA	NA
>>=	>>	NA	NA	NA	NA	NA	NA	NA
&=	bitwise &	NA	NA	logical &	NA	NA	NA	NA
=	bitwise	NA	NA	logical	NA	concat	NA	NA
^=	bitwise ^	NA	NA	logical ^	NA	NA	NA	NA

The scope of a variable depends on the location where the variable is assigned the first time. For instance, in the following code:

```
let a := 1
foreach task in TASKS do
  let b := 2
  let a += 1
end foreach
println a
println b
```

Because a is assigned for the first time outside the **foreach** loop, it is both accessible within the **foreach** loop and accessible after the **foreach** loop. So it contains the number of items in TASKS + 1 after the **foreach**. Because b is assigned for the first time inside the **foreach** loop, its scope is set within the **foreach** loop and it does not exist after the loop anymore and **println** b will trigger and error.

3.3 The unlet instruction

The unlet instruction removes a variable, a struct field, a map item or a list item. The variable / struct field / map item / list item ceases to exist. If the variable / struct field / map item / list does not exist, unlet fails silently. Here are some examples.

```
let a := 0
if exists a then
    println "'a' found"
else
    println "'a' not found"
end if
unlet a
if exists a then
    println "'a' found"
```

```
else
  println "'a' not found"
end if
```

```
'a' found
'a' not found
```

Example 2

Here unlet is used to remove a field from a struct:

```
let myStruct := @{ a: 1, b: 2, c: 3 }
unlet myStruct::a
display myStruct
```

```
myStruct - struct: @{
    b :>
        integer: 2
    c :>
        integer: 3
}
```

Example 3

Here we use unlet to remove an item from a list:

```
let myList := @( 1, 2, 3, 4 )
unlet myList[2]
display myList
```

```
myList - list: @(
    0 :>
        integer: 1
    1 :>
        integer: 2
    2 :>
        integer: 4
)
```

Example 4

And here to remove an item from a map

```
let myMap := @[ "a": @( 1, 2) , "b": @( 3, 4) ]
unlet myMap["b"]
display myMap
```

3.4 The! instruction

The ! instruction emits an expression in the output template string. The syntax is

```
! expression
```

For instance the following program:

```
loop i from 1 to 10 do
    !" " !i
end loop
```

```
1 2 3 4 5 6 7 8 9 10
```

3.5 The ? instruction

The ? instruction get the current column index in the output string.

```
? var
```

Used with the tab instruction, see 3.6, ? allows flexible formatting.

3.6 The tab instruction

The **tab** instruction emits spaces in the output string until the column given in argument is reached. If the current column is greater or equal than the column given in argument, no space is emitted.

```
tab expression
```

expression should be an int expression, otherwise a runtime error occurs.

3.7 The sort instruction

The **sort** instruction sorts a list. If elements of the list support operators < and > they are sorted using these two operators.

```
sort var > # descending order
sort var < # ascending order</pre>
```

```
let aList := @( "wish", "you", "where", "here" )
sort aList >
display aList
sort aList <
display aList</pre>
```

```
aList from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 17:7
 - list: @(
   0 :>
       string: "you"
   1 :>
       string: "wish"
       string: "where"
       string: "here"
aList from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 19:7
 - list: @(
   0 :>
       string: "here"
   1 :>
       string: "where"
   2 :>
       string: "wish"
       string: "you"
```

If the elements of the list are struct, a second form exists:

```
sort var by identifier < # descending order
sort var by identifier < # ascending order</pre>
```

In this case, the field *identifier* of the list item is used as sorting key.

```
aList from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 27:7
  - list: @(
   0 :>
       struct: @{
           age :>
               integer: 29
           height :>
               integer: 175
           name :>
              string: "John"
       }
   1 :>
       struct: @{
           age :>
               integer: 22
           height :>
              integer: 170
           name :>
```

```
string: "Bob"
   2 :>
       struct: @{
           age :>
               integer: 18
           height :>
               integer: 180
           name :>
               string: "Arnold"
aList from file '/Users/jlb/Develop/GTL/examples/mapofTest.gtl', line 29:7
 - list: @(
   0 :>
       struct: @{
           age :>
               integer: 22
           height :>
              integer: 170
           name :>
               string: "Bob"
   1 :>
       struct: @{
           age :>
               integer: 29
           height :>
              integer: 175
           name :>
               string: "John"
       }
   2 :>
       struct: @{
           age :>
               integer: 18
           height :>
              integer: 180
           name :>
               string: "Arnold"
       }
```

3.8 The if instruction

if is the conditional execution instruction. The forms are:

```
if expression then
   instruction_list
end if

if expression then
   instruction_list
else
   instruction_list
end if

if expression then
   instruction_list
```

```
elsif expression then
  instruction_list
end if

if expression then
  instruction_list
elsif expression then
  instruction_list
else
  instruction_list
end if
```

The expression must be a bool. In the following example, the blue text (within the '%') is produced only if the USECOM bool variable is true:

```
if USECOM then %
#include "tpl_com.h" %
end if
```

3.9 The foreach instruction

This instruction iterates on the elements of a collection, a list or a map. The simplest form is the following one:

```
foreach var in expression do
  instruction_list
end foreach
```

Here var takes the value of each of the elements of the collection. If the collection is a list, the elements are iterated in the order of the list. If the collection is a map, the element are iterated in the alphanumerical order of the keys. In both cases, a variable named INDEX which contains the current iteration number is available inside the loop. INDEX ranges from 0 to the number of elements in the list minus 1. If the collection is a map, a second variable, KEY, which contains the key associated to the value of the current item, is available.

In the following example, for each element in the ALARMS list, the text between the do and the end foreach is produced with the NAME attribute of the current element of the ALARMS list inserted at the specified location.

```
foreach alr in ALARMS do
%
/* Alarm % !alr::NAME % identifier */
#define % !alr::NAME %_id % !INDEX %
CONST(AlarmType, AUTOMATIC) % !alr::NAME % = % !NAME %_id;
%
end foreach
```

A more general form of the foreach instruction is:

```
foreach key, var (index_var) in expression
before
   instruction_list
do
   instruction_list
between
   instruction_list
after
```

```
instruction_list
end foreach
```

key may be used only when iterating on a map and allows to give a custom name to the default key variable. (index_var) may be used both for a list or a map and allows to give a custom name to the default INDEX variable.

If the collection is not empty, the **before** section is executed once before the first execution of the **do** section. If the collection contains at least two elements, the **between** section is executed between the execution of the **do** section. If the list is not empty, the **after** section is executed once after the last execution of the **do** section.

The following example illustrates the general form. Here a table of pointers to alarm descriptors is generated:

```
#define ALARM_COUNT 2
tpl_time_obj *tpl_alarm_table[ALARM_COUNT] = {
   &alr1_alarm_desc,
   &alr2_alarm_desc
};
```

3.10 The for instruction

The **for** instruction iterates along a literal list of elements.

```
for var in expression, ..., expression do
   ...
end for
```

At each iteration, var gets the value of the current expression. As in the foreach instruction, INDEX is generated and ranges from 0 to the number of elements in the list minus 1.

The for instruction is deprecated. Use foreach with a literal list instead.

3.11 The loop instruction

The loop instruction iterates over a range of integers. Its simplest form is:

```
loop var from expression_start to expression_end do
   ...
end loop
```

Both expression_start and expression_end must be integer expressions. By default var is incremented by one from expression_start, inclusive, to expression_end, inclusive.

Like in the foreach instruction, **before**, **between** and **after** sections may be used. Moreover, **down** may be used to decrement *var* by one. **up** is a syntactic sugar which is here for symmetry purpose and may be omitted. **step** allows to increment or decrement by *increment*. If **step** is omitted, **step** 1 is assumed.

```
loop var from expression <up | down> to expression <step increment>
  before ...
  do ...
  between ...
  after ...
end loop
```

For instance, in the following loop, a goes from 0 to 10 with an increment of 2:

```
loop a from 0 to 10 step 2 do
  println a
end loop
```

```
0
2
4
6
8
10
```

In the following loop, a goes from 25 to 20 with a decrement of 1:

```
loop a from 25 down to 20 do
  println a
end loop
```

```
25
24
23
22
21
20
```

Because the step can be a negative integer number, this output may be produced by the following program too:

```
loop a from 25 to 20 step -1 do
  display a
end loop
```

 \wedge

Despite the use of big integers the number of iteration is limited to $2^{32} - 1$




Figure 3.1: semantics of the repeat instruction

3.12 The repeat instruction

The repeat instruction combines the C while (...) { ... } and do { ... } while (...); in one instruction. The general form is:

```
repeat <(limit_expression)>
  instruction_list_1
while condition do
  instruction_list_2
end repeat
```

 $limit_expression$ is an optional expression to bound the number of iterations. If the number of iterations exceeds $limit_expression$, a runtime error is emitted. If $limit_expression$ is omitted, its default value is $2^{32}-1$.

The semantics of this instruction is shown at figure 3.1

```
let aList := @( 1, 2, 3, 4 )
repeat ( [aList length] )
while [aList length] > 0 do
  println [aList first]
  unlet aList[0]
end repeat
```

```
3
```

3.13 The write instruction

The write instruction writes the template output string to a file. The general form is:

```
write to <executable> expression :
   instruction_list
end write
```

Where <code>expression</code> is a string expression. Here the template output string built by instructions in the <code>instruction_list</code> are written to the file named after the evaluation of <code>expression</code>. If the optional keyword <code>executable</code> is present, the file has its executable bit set.

Example 1

```
write to "/tmp/outputOfTemplate" :
%Hello
%
end write

Created '/tmp/outputOfTemplate'.
```

The content of file at path '/tmp/outputOfTemplate' is:

```
Hello
```

Example 2

```
write to executable "/tmp/listfiles.sh" :
%# /bin/sh
ls -al
%
end write
Created '/tmp/listfiles.sh'.
```

The content of file at path '/tmp/outputOfTemplate' is:

```
# /bin/sh
ls -al
```

And since it is an executable file, one can type '/tmp/listfiles.sh' to execute the shell script.

3.14 The template instruction

The **template** instruction invokes another template. The output string built by the invoked template is included in the output string of the caller at the location occupied by the **template** instruction. The general form is:

3.14.1 Template invocation with a copy of variables

The simplest form is:

```
template file_reference
```

The file_reference may be an identifier or from followed by an expression that evaluate to a string.

```
template aTemplate # identifier file reference
```

 α r

```
template from "aTemplate" # string file reference
```

The second one allow to use any file name. In both cases, the **template** instruction looks for a file named <code>aTemplate.extension</code>. <code>extension</code> is customizable. For gtl it is set to <code>"gtl"</code>. For goil it is set to <code>"goilTemplate"</code>.

If the template file is found it get a copy of the variables of the caller and is executed. If it is not found a runtime error occurs.

Example

Contents of callerTemplate.gtl:

```
loop a from 1 to 4 do
  template helloTemplate
  % % !a % %
end loop
%
```

Contents of helloTemplate.gtl:

```
Hello
```

```
Hello 1 Hello 2 Hello 3 Hello 4
```

3.14.2 Template invocation with arguments

Instead of passing a copy of all the variables to the invoked template, it is possible to pass arguments. In this case the invoked template works as a procedure and has no access to the variables of the caller. See 3.15 too.

Example

Contents of caller2Template.gtl:

```
loop a from 1 to 4 do
  template (a) hello2Template
end loop
%
```

Contents of hello2Template.gtl:

```
input (number)
%Hello % !number % %
```

3.14.3 Conditional invocation of templates

In some cases, the invocation of a template may be optional and if the template is not found, no error should occurs. The following form goes this:

```
template if exists file_reference
```

If the template file is found it get a copy of the variables of the caller and is executed. If it is not found the template invocation fails silently.

When the template file is not found it is possible to execute a list of instructions. This is done by adding a or, the instruction list and end template.

```
template if exists file_reference or
  instruction_list
end template
```

3.15 The input instruction

The **input** instruction is used in a template to retrieve the arguments passed to the template by the caller. See section 3.14.2

```
input (formal_argument_list)
```

input may be used at any time in the template. For each variable appearing in the formal_argument_list input pops the first value from the argument list passed by the caller. This can be done by one or more input instructions.

Example

```
# The caller invoke a template with argument 1, 2, 3 and 4
template (1, 2, 3, 4) aTemplate
```

Contents of file aTemplate.gtl:

```
input(a) # retrieve 1
input(b, c) # retrieve 2 in b and 3 in c
input(d) # retrieve 4
input(e) # trigger a runtime error, the argument list is empty
```

Arguments in the <code>formal_argument_list</code> may be typed. In the following example any data type may be passed for <code>d</code> but <code>c</code> requires an int and <code>a</code> requires a string.

Example

```
template (3, 2, 1) aTemplate
```

Contents of file aTemplate.gtl:

```
input(c : @int, d) # retrieve 3 in c and 2 in d
input(a : @string) # trigger an error int 1 is not a string
```

3.16 The error and warning instructions

It can be useful to generate an error or a warning if a data is not defined or if its value is unappropriate. error and warning have 2 forms:

```
error var : expression
warning var : expression
```

or

```
error here : expression
warning here : expression
```

expression must be of type string. In the first form, var is a variable. The file location of this variable may be a location in any input file of the compiler which embeds the GTL interpreter or in the template file if the variable was assigned in the template. This location is used to signal the location of the error or warning. In the second form, here means the current location in the template file.

In the following example taken from the Goil templates, an error is generated if the ACTIVATION attribute of an extended task is greater than $\boxed{1}$:

Example 1

```
# Check no extended task as an ACTIVATION attribute greater than 1
foreach task in EXTENDEDTASKS do
  if task::ACTIVATION > 1 then
    error task::ACTIVATION : "An extended task cannot have ACTIVATION greater than 1"
  end if
end foreach
```

In this second example, a warning is generated if a template is not found:

Example 2

```
template if exists interrupt_wrapping or
  warning here : "interrupt_wrapping.goilTemplate not found"
end template
```

3.17 The print and println instructions

print an println print an expression to the standard output. println prints a '\n' after the expression. println may be used alone to print a '\n'.

```
print expression
println <expression>
```

Any string, int, bool, float, enum, type and char may be printed. struct, list, map and unconstructed may not.

```
foreach a in @( "Does", "anybody", "remember", "Vera", "Lynn", '?' ) do
  print a
  print " "
end foreach
println
```

3.18 The display instruction

display prints the name of any variable, the location of the display instruction and the content of any variable. It is designed to be use for debug purpose. To get the following output, a display TASKS has been added to root.goilTemplate and the example in examples/cortex/-armv7em/stm32f407/stm32f4discovery/alarms has been compiled by Goil.

```
TASKS from file '/Users/jlb/Develop/trampoline-git-maintain/goil/templates/root.
   goilTemplate', line 1467:7
  - list: @(
   0 :>
        struct: @{
           ACTIVATION :>
                integer: 1
            AUTOSTART :>
               boolean: false
            KIND :>
               string: "Task"
            NAME :>
               string: "blink"
            NONPREEMPTABLE :>
               boolean: false
            PRIORITY :>
               integer:
            SCHEDULE :>
               string: "FULL"
            STACKSIZE :>
               integer: 300
            USEFLOAT :>
               boolean: false
            USEINTERNALRESOURCE :>
               boolean: false
   1 :>
        struct: @{
           ACTIVATION :>
               integer: 1
            AUTOSTART :>
               boolean: false
            KIND :>
               string: "Task"
            NAME :>
               string: "read_button"
            NONPREEMPTABLE :>
               boolean: false
            PRIORITY :>
               integer: 2
            SCHEDULE :>
               string: "FULL"
            STACKSIZE :>
               integer: 300
            USEFLOAT :>
                boolean: false
            USEINTERNALRESOURCE :>
```

```
boolean: false
}
)
```

3.19 The variables instruction

variables displays all the variables. It is designed to be use for debug purpose.

```
variables
```

```
let a := @( 1, 2, 3 )
let b := @{ x:1, y:2, z:3 }
let c := @[ "age": 10, "name": "Vera" ]
let d := "Hello"
let e := 3
variables
```

```
======= Displayed from ==
file '/Users/jlb/Develop/GTL/examples/variables.gtl', line 7:9
______
list: @(
  0 :>
     integer: 1
  1 :>
     integer: 2
     integer: 3
b
struct: @{
  x :>
     integer: 1
  у :>
     integer: 2
  z :>
     integer: 3
map: @[
  "age" :>
    integer: 10
  "name" :>
     string: "Vera"
string: "Hello"
```

e _____integer: 3 _____

Chapter 4

GTL modules

GTL may be extended with functions, setters and getters. Definitions of these objects are done in separates <code>.gtm</code> files called modules.

4.1 Importing a module

Modules are imported by using the **import** statement. GTL prevents multiple import of the same module.

```
import expression
```

expression must evaluate to a string. import is not an instruction and must appear at the beginning of the template file before any instruction except the \$...\$ instruction.

4.2 Writing a module

A module includes zero or more functions definitions, zero or more getter definitions and zero or more setter definitions. It may include <code>import</code> statements too but they must appear before any function / getter / setter definition. If <code>%...%</code> instructions are used they are ignored. Other instructions which output data in the output template string are forbidden.

4.2.1 The arguments

Arguments are passed by copy. Each formal argument of the list may be typed or not. If a formal argument is typed, GTL emit a runtime error if the passed argument does not have the same type. Here is a formal argument list with the second argument typed.

```
a, b : @int, c
```

Any data type may be passed for arguments a and c but an int is required for argument

The type of an argument may be tested, check 2.15, 2.3.2 and 4.2.2.

4.2.2 Function definition

A function definition has the following form:

```
func func_name(formal_argument_list) result_var
  instruction_list
end func
```

The <code>formal_argument_list</code> may be empty. <code>result_var</code> is a local variable of the function which is used to return the result.

Example 1

Content of function.gtm:

```
func square(x) result
let result := x * x
end func
```

Content of template.gtl:

```
import "function"

let b := square(6)
display b
let b := square(2.6)
display b
```

```
b from file '/Users/jlb/Develop/GTL/examples/template.gtl', line 5:7
  - integer: 36
b from file '/Users/jlb/Develop/GTL/examples/template.gtl', line 7:7
  - float: 6.76
```

Example 2

Content of function.gtm:

```
func square(x : @int) result
  let result := x * x
end func
```

Content of template.gtl:

```
import "function"

let b := square(6)
display b
let b := square(2.6)
display b
```

```
b from file '/Users/jlb/Develop/GTL/examples/template.gtl', line 5:7
  - integer: 36
/Users/jlb/Develop/GTL/examples/template.gtl:6:17:19:
semantic error #1: int expected for x
let b := square(2.6)
______^^^
```

Example 3

Content of function.gtm:

```
func square(x) result
  if [x isANumber] then
   let result := x * x
  else
    error here : "int or float expected"
  end if
end func
```

Content of template.gtl:

```
import "function"

let b := square(6)
display b
let b := square(2.6)
display b
let b := square("Hello")
display b

b from file '/Users/jlb/Develop/GTL/examples/template.gtl', line 5:7
    - integer: 36
b from file '/Users/jlb/Develop/GTL/examples/template.gtl', line 7:7
    - float: 6.76
/Users/jlb/Develop/GTL/examples/template.gtl:6:18:24:
semantic error #1: int or float expected
let b := psquare("Hello")
```

4.3 Getter definition

A getter is defined as follow:

```
getter a_type getter_name(formal_argument_list) result_var
  instruction_list
end getter
```

a_type designates the data type on which the getter apply. It can be @int , @char , @float , @bool , @enum , @string , @struct , @list , @map , @type or @unconstructed . getter_name is the name of the getter. The same name may be used for several types. The formal_argument_list follows the same rules as those exposed in 4.2.2. result_var is the name of the variable used to store the result of the getter.

Within the getter, variable **self** is a copy of the variable targeted by the getter. It can me modified but the target will not be touched.

Example 1

In this example a getter to get the square of an int is defined Content of getter.gtm:

```
# getter : square of a int
#-----
getter @int square() result
  let result := self * self
end getter
```

Content of template.gtl:

```
import "getter"

let b := [6 square]
display b
let b := [b square]
display b
```

```
b from file '/Users/jlb/Develop/GTL/examples/testGetter.gtl', line 5:7
  - integer: 36
b from file '/Users/jlb/Develop/GTL/examples/testGetter.gtl', line 7:7
  - integer: 1296
```

Example 2

In this example two getters, one to check a struct and the other to check a list are defined Content of getter.gtm:

Content of template.gtl:

```
List 1 ok
List 2 ko
```

4.4 Setter definition

A setter is defined as follow:

```
setter a_type setter_name(formal_argument_list)
  instruction_list
end getter
```

a_type designates the data type on which the setter apply. It can be @int , @char , @float , @bool , @enum , @string , @struct , @list , @map , @type or @unconstructed . setter_name is the name of the setter. The same name may be used for several types. The formal_argument_list follows the same rules as those exposed in 4.2.2.

Within the setter, variable self is the variable targeted by the setter.

Example

Content of setter.gtm:

Content of template.gtl:

```
let c := @(
    @{ age : 18, height : 180, name : "Arnold" },
    @{ age : 22, height : 170, name : "Bob"            },
    @{ age : 29, height : 175, name : "John"            },
    @{                        height : 150, name : "Sally"            }
)

[!c deleteAge]
display c
```