

# GTL Template language

Jean-Luc Béchenec

July 30, 2016

## Contents

<b>1</b>	<b>Data types</b>	<b>2</b>
1.1	getters applicable to any data type . . . . .	2
1.2	int data type . . . . .	3
1.2.1	int operators . . . . .	3
1.2.2	int getters . . . . .	4
1.2.3	int setters . . . . .	5
1.3	The float data type . . . . .	5
1.3.1	float operators . . . . .	5
1.3.2	float getters . . . . .	6
1.4	The string data type . . . . .	6
1.4.1	string operators . . . . .	7
1.4.2	string getters . . . . .	7
1.5	The bool data type . . . . .	10
1.5.1	bool operators . . . . .	10
1.5.2	bool getters . . . . .	11
1.6	The struct data type . . . . .	11
1.6.1	struct operators . . . . .	11
1.6.2	struct getter . . . . .	12
1.7	The list data type . . . . .	12
1.7.1	list operators . . . . .	12
1.7.2	list getters . . . . .	12
1.8	The map data type . . . . .	14
1.8.1	map operators . . . . .	14
<b>2</b>	<b>GTL instructions</b>	<b>15</b>
2.1	The <i>let</i> instruction . . . . .	15
2.2	The <i>if</i> instruction . . . . .	15
2.3	The <i>foreach</i> instruction . . . . .	16
2.4	The <i>for</i> instruction . . . . .	17
2.5	The <i>loop</i> instruction . . . . .	17

# 1 Data types

GTL supports the following data types:

**int** arbitrary precision integer numbers. The GMP library is used;

**float** 64 bits floating point numbers;

**bool** standard boolean;

**type** the type of a data;

**string** unicode strings;

**struct** structured data;

**list** lists of data, may be accessed as a table;

**map** map (aka dictionary) of data;

**unconstructed** an unconstructed variable.

Each type has its set of operators, getters and setters. The expression, for getters, or the variable, for setters, is called the *target*. Getters return a value related to the data but do not change it. They are used to get information from a data or to convert it into another type. Setter may target a literal expressions. Setters change the data and do not return anything. Setter may only target a variable. Getters and setters may have arguments. Syntax for getters without argument is as follow:

`[expression getter]`

When the getter takes arguments, they are listed after a colon and separated commas as follow:

`[expression getter : arg1, arg2, ..., argN]`

Syntax for setters without arguments is as follow:

`[!variable setter]`

When the setter takes arguments, they are listed after a colon and separated by commas as follow:

`[!variable setter : arg1, arg2, ..., argN]`

## 1.1 getters applicable to any data type

**type** returns the data type of the expression. See the section ??.

**isANumber** returns `true` if the expression is a number: `int` or `float`, `false` otherwise.

## 1.2 int data type

The int data type support arbitrary precision arithmetic by using the GNU Multiple Precision Arithmetic Library (GMP).

### 1.2.1 int operators

The int datatype supports the following operators:

#### Unary operators

Operator	Expression type	Meaning
+	int $\leftarrow$ +int	Plus operator. No effect
-	int $\leftarrow$ -int	Minus operator. Negation
~	int $\leftarrow$ ~int	Not operator. Complementation by 1

#### Binary arithmetic operators

Operator	Expression type	Meaning
+	int $\leftarrow$ int + int	Addition
-	int $\leftarrow$ int - int	Substraction
*	int $\leftarrow$ int * int	Multiplication
/	int $\leftarrow$ int / int	Division
mod	int $\leftarrow$ int mod int	Modulus

#### Binary bitwise operators

Operator	Expression type	Meaning
&	int $\leftarrow$ int & int	bitwise and
	int $\leftarrow$ int   int	bitwise or
^	int $\leftarrow$ int ^ int	bitwise exclusive or
<<	int $\leftarrow$ int << int	shift left
>>	int $\leftarrow$ int >> int	shift right

#### Comparison operators

Operator	Expression type	Meaning
!=	bool $\leftarrow$ int != int	Not equal
==	bool $\leftarrow$ int == int	Equal
>	bool $\leftarrow$ int > int	Greater than
<	bool $\leftarrow$ int < int	Lower than
>=	bool $\leftarrow$ int >= int	Greater or equal
<=	bool $\leftarrow$ int <= int	Lower or equal

### 1.2.2 int getters

getter	Type	Meaning
string	string	Returns a string decimal representation of the int expression. [42 string] returns string "42".
hexString	string	Returns a string hexadecimal representation of the int expression prefixed by 0x. If the expression is negative a '-' is inserted before. [42 hexString] returns string "0x2A". [-1 hexString] returns string "-0x1".
xString	string	Returns a string hexadecimal representation of the int expression. If the expression is negative a '-' is inserted before. [42 xString] returns string "2A". [-42 xString] returns string "-2A".
numberOfBytes	int	Returns the number of bytes needed to store an unsigned expression. [255 numberOfBytes] returns 1, [256 numberOfBytes] returns 2
signedNumberOfBytes	int	Returns the number of bytes needed to store a signed expression. [127 numberOfBytes] returns 1, [128 numberOfBytes] returns 2
numberOfBits	int	Returns the number of bits needed to store an unsigned expression. [63 numberOfBits] returns 6, [64 numberOfBits] returns 7
signedNumberOfBits	int	Returns the number of bits needed to store a signed expression. [63 signedNumberOfBits] returns 7, [64 signedNumberOfBits] returns 8
sign	int	Returns -1 if the expression is strictly negative, 0 if it is null and +1 if the expression is strictly positive.
fitsUnsignedInByte	bool	Returns true if the expression fits in an unsigned byte, false otherwise.
fitsSignedInByte	bool	Returns true if the expression fits in a signed byte, false otherwise.
fitsUnsignedInWord	bool	Returns true if the expression fits in an unsigned 16 bits word, false otherwise.
fitsSignedInWord	bool	Returns true if the expression fits in a signed 16 bits word, false otherwise.
fitsUnsignedInLong	bool	Returns true if the expression fits in an unsigned 32 bits long, false otherwise.
fitsSignedInLong	bool	Returns true if the expression fits in a signed 32 bits long, false otherwise.

getter	Type	Meaning
<code>fitsUnsignedInLongLong</code>	bool	Returns <b>true</b> if the expression fits in an unsigned 64 bits long long, <b>false</b> otherwise.
<code>fitsSignedInLongLong</code>	bool	Returns <b>true</b> if the expression fits in a signed 64 bits long long, <b>false</b> otherwise.
<code>abs</code>	int	Returns the absolute value of the expression.
<code>bitAtIndex</code>	bool	This getter takes one argument: <code>index</code> . It returns <b>true</b> if the bit at <code>index</code> is set and <b>false</b> otherwise. <code>index</code> 0 corresponds to the lowest significant bit. [1 <code>bitAtIndex: 0</code> ] returns <b>true</b>

### 1.2.3 int setters

setter	Meaning
<code>setBitAtIndex</code>	This setter takes two arguments. The first one, <code>value</code> , is a bool. The second one, <code>index</code> , is the index of the bit to set. if <code>value</code> is <b>true</b> the bit is set to 1 and to 0 otherwise. Assuming <code>a</code> contains 0 at start, [ <code>!a setBitAtIndex: true, 0</code> ] sets <code>a</code> to 1.
<code>complementBitAtIndex</code>	This setter takes one argument, <code>index</code> , which is the index of the bit to complement. Assuming <code>a</code> contains 1 at start, [ <code>!a complementBitAtIndex: 1</code> ] sets <code>a</code> to 3.

## 1.3 The float data type

The float data type is the standard IEEE754 64 bits floating point number.

### 1.3.1 float operators

The float data type supports the following operators:

#### Unary operators

Operator	Expression type	Meaning
<code>+</code>	<code>float ← +float</code>	Plus operator. No effect
<code>-</code>	<code>float ← -float</code>	Minus operator. Negation

#### Binary arithmetic operators

Operator	Expression type	Meaning
<code>+</code>	<code>float ← float + float</code>	Addition
<code>-</code>	<code>float ← float - float</code>	Substraction
<code>*</code>	<code>float ← float * float</code>	Multiplication

Operator	Expression type	Meaning
/	float $\leftarrow$ float / float	Division

### Comparison operators

Operator	Expression type	Meaning
!=	bool $\leftarrow$ float != float	Not equal
==	bool $\leftarrow$ float == float	Equal
>	bool $\leftarrow$ float > float	Greater than
<	bool $\leftarrow$ float < float	Lower than
>=	bool $\leftarrow$ float >= float	Greater or equal
<=	bool $\leftarrow$ float <= float	Lower or equal

### 1.3.2 float getters

getter	Type	Meaning
string	string	Returns a string representation of the float expression. [4.2 string] returns string "4.2".
cos	float	Returns the cosine of a float expression expressed in radian.
sin	float	Returns the sine of a float expression expressed in radian.
tan	float	Returns the tangent of a float expression expressed in radian.
cosDegree	float	Returns the cosine of a float expression expressed in degree.
sinDegree	float	Returns the sine of a float expression expressed in degree.
tanDegree	float	Returns the tangent of a float expression expressed in degree.
exp	float	Returns the exponentiation of a float expression.
logn	float	Returns the natural logarithm of a float expression.
log2	float	Returns the logarithm base 2 of a float expression.
log10	float	Returns the logarithm base 10 of a float expression.
sqrt	float	Returns the square root of a float expression.
power	float	This getter takes one argument, p. It returns the expression raised to the power of p.

## 1.4 The string data type

The **string** data type supports unicode. A literal string is delimited by a pair of ". Literal strings support special characters:

Escape sequence	Corresponding character
\f	form feed
\n	new line
\r	return
\t	horizontal tab
\v	vertical tab

Escape sequence	Corresponding character
<code>\\</code>	backslash
<code>\0</code>	null character
<code>\unnnn</code>	unicode character with code <i>nnnn</i> in hexadecimal
<code>\Unnnnnnnn</code>	unicode character with code <i>nnnnnnnn</i> in hexadecimal

#### 1.4.1 string operators

The `string` data type supports the following operators:

##### Binary operator

Operator	Expression type	Meaning
<code>+</code>	<code>string ← string + string</code>	Concatenation

##### Comparison operators

Operator	Expression type	Meaning
<code>!=</code>	<code>bool ← string != string</code>	Not equal
<code>==</code>	<code>bool ← string == string</code>	Equal
<code>&gt;</code>	<code>bool ← string &gt; string</code>	Greater than
<code>&lt;</code>	<code>bool ← string &lt; string</code>	Lower than
<code>&gt;=</code>	<code>bool ← string &gt;= string</code>	Greater or equal
<code>&lt;=</code>	<code>bool ← string &lt;= string</code>	Lower or equal

#### 1.4.2 string getters

getter	Type	Meaning
<code>HTMLRepresentation</code>	<code>string</code>	Returns a representation of the string suitable for an HTML encoded representation. ‘&’ is encoded by <code>&amp;amp;</code> ; ‘”’ by <code>&amp;quot;</code> ; ‘<’ by <code>&amp;lt;</code> ; and ‘>’ by <code>&amp;gt;</code> .

getter	Type	Meaning
identifierRepresentation	string	Returns an unique representation of the string conforming to a C identifier. Any Unicode character that is not a latin letter is transformed into its hexadecimal code point value, enclosed by ‘_’ characters. This representation is unique: two different strings are transformed into different C identifiers. For example: <code>value3</code> is transformed to <code>value_33_</code> ; <code>+=</code> is transformed to <code>_2B__3D_</code> ; <code>An_Identifier</code> is transformed to <code>An_5F_Identifier</code> .
fileExists	bool	Returns <code>true</code> if a file exists at the target path, <code>false</code> otherwise.
length	integer	Returns the number of characters in the string
lowercaseString	string	Returns the lowercased representation of the string.
capitalized	string	if the string is empty, this getter returns the empty string; otherwise, it returns the string with the first character being replaced with the corresponding upper case character.
uppercaseString	string	Returns uppercased representation of the receiver
leftSubString	string	Returns the sub-string from the beginning of the target and with the number of characters passed as argument. If the sub-string is longer than the target, the target is returned. <code>["Hello_World_!" leftSubString : 5]</code> returns <code>"Hello"</code> .
rightSubString	string	Returns the sub-string from the end of the target and with the number of characters passed as argument. If the sub-string is longer than the target, the target is returned. <code>["Hello_World_!" leftSubString : 11] rightSubString: 5]</code> returns <code>"World"</code> .



getter	Type	Meaning
subString	string	Returns the sub-string from the <code>index</code> passed as first argument and with the number of characters passed as second argument. If the <code>index</code> is out of the target, the empty string is returned. If the number of characters is greater than the sub-string, the sub-string is returned. <code>["Hello_World_!" subString : 6, 5]</code> returns "World". <code>["Hello" subString : 10, 3]</code> returns the empty string. <code>["Hello" subString : 2, 10]</code> returns "llo".
reversedString	string	Returns a mirrored string. <code>["Hello_World_!" reversedString]</code> returns "!_dlroW_olleH".
componentsSeparatedByString	list	This getter takes one string argument: <code>separator</code> . The target is cut into pieces according to the separator and a list of the pieces is returned. <code>["Hello_World_!" componentsSeparatedByString : " "]</code> returns <code>@( "Hello", "World", "!" )</code> .
columnPrefixedBy	string	This getter takes one string argument: <code>prefix</code> . Return the target with each line prefixed by <code>prefix</code> . <code>["Hello\nWorld" columnPrefixedBy : "#_"]</code> returns <code>"#_Hello\n#_World"</code> .
wrap	string	Wraps the target to a width. This getter takes two int arguments: <code>width</code> and <code>shift</code> . The target is assumed to contain paragraphs separated by <code>\n</code> . Returns the target with each paragraph wrapped to <code>width</code> . In addition, each line of the paragraph except the first one is prefixed by <code>shift</code> spaces. <code>["Hello_beautiful_World.\nHow_are_you" wrap : 6, 2]</code> returns <code>"Hello\n _beautiful\n _World\nHow\n _are\n you"</code> .

getter	Type	Meaning
subStringExists	bool	This getter takes one argument, <code>subString</code> . It returns <code>true</code> if the sub-string <code>subString</code> is found in the target, <code>false</code> otherwise.
replaceString	string	This getter takes two argument, <code>find</code> and <code>replace</code> . It returns the target where each occurrence of <code>find</code> is replaced by <code>replace</code> .
envVar	string	Returns the value of the target environment variable. If it does not exists, <code>envVar</code> returns the empty string.
envVarExists	bool	Returns <code>true</code> if target environment variable exists, <code>false</code> otherwise.

## 1.5 The `bool` data type

A true literal bool can be written as `true` or `yes` and a false literal bool can be written as `false` or `no`.

### 1.5.1 `bool` operators

The `bool` data type supports the following operators:

#### Unary operator

Operator	Expression type	Meaning
<code>~</code>	<code>bool ← bool</code>	logical not

#### Binary operator

Operator	Expression type	Meaning
<code>&amp;</code>	<code>bool ← bool &amp; bool</code>	logical and
<code> </code>	<code>bool ← bool   bool</code>	logical or
<code>^</code>	<code>bool ← bool ^ bool</code>	logical exclusive or

#### Comparison operators

For comparison operators, `false` is considered to be lower than `true`.

Operator	Expression type	Meaning
<code>!=</code>	<code>bool ← bool != bool</code>	Not equal
<code>==</code>	<code>bool ← bool == bool</code>	Equal
<code>&gt;</code>	<code>bool ← bool &gt; bool</code>	Greater than

Operator	Expression type	Meaning
<	<code>bool ← bool &lt; bool</code>	Lower than
>=	<code>bool ← bool &gt;= bool</code>	Greater or equal
<=	<code>bool ← bool &lt;= bool</code>	Lower or equal

### 1.5.2 bool getters

getter	Type	Meaning
<code>trueOrFalse</code>	string	Returns a string representation, "true" or "false" of the bool expression.
<code>string</code>	string	Returns a string representation, "true" or "false" of the bool expression.
<code>yesOrNo</code>	string	Returns a string representation, "yes" or "no" of the bool expression.
<code>TRUEOrFALSE</code>	string	Returns a string representation, "TRUE" or "FALSE" of the bool expression.
<code>YESOrNO</code>	string	Returns a string representation, "YES" or "NO" of the bool expression.
<code>int</code>	int	Returns an int representation, 1 or 0 of the bool expression.

## 1.6 The struct data type

The struct data type allows to store a heterogeneous set of data in one variable. Struct members are accessed by using the `::` separator. If *A* is a struct, *A::B* refers to field *B* of *A*.

A literal struct is defined as follow:

```
@{ a: 1, b: 2, c: 3 }
```

This define a struct with fields a, b and c and respective values 1, 2 and 3.

### 1.6.1 struct operators

The struct data type supports the following operators:

Operator	Expression type	Meaning
<code>!=</code>	<code>bool ← struct != struct</code>	Not equal
<code>==</code>	<code>bool ← struct == struct</code>	Equal

Two structs are equal if:

- they have the same number of field
- they have the same field names
- they have the same field values

### 1.6.2 struct getter

getter	Type	Meaning
map	map	Returns a map representation.

## 1.7 The list data type


The list data type allows to store a list of data. list items are accessed by using `[<number>]` where `<number>` is the rank of the element starting at 0. If `A` is a list, `A[0]` refers to element 0 of `A`.

A literal list is defined as follow:

```
@( 1, 2, 3 )
```

This define a list of int with elements 1, 2 and 3.

An empty list can be initialized using the `emptylist` constant.

 The `emptylist` constant is deprecated. Use a literal empty list, `@()`, instead.

### 1.7.1 list operators

The list data type supports the following operators:

#### Binary operators

Operator	Expression type	Meaning
+	<code>list ← list + any</code>	add <code>any</code> at the end of the list
	<code>list ← list   list</code>	Concatenate lists

#### Comparison operators

Operator	Expression type	Meaning
!=	<code>bool ← list != list</code>	Not equal
==	<code>bool ← list == list</code>	Equal

Two structs are equal if:

- they have the same number of elements
- they have the same elements values

### 1.7.2 list getters

getter	Type	Meaning
length	int	Returns the number of elements in the list.

getter	Type	Meaning
first	any	Returns the first element of the list.
last	any	Returns the last element of the list.
mapBy	map	<b>mapBy</b> takes on string argument which is the field (for a struct list item) or the key (for a map list item) used as key to store the element in the resulting map. It returns a map where each element is the element of the list with the key being the corresponding field/key.

### example of mapBy

The following code snippet:

```
let myList := @(
  @{
    age : 18,
    height : 180,
    name : "Arnold"
  },
  @{
    age : 22,
    height : 170,
    name : "Bob"
  },
  @{
    age : 29,
    height : 175,
    name : "John"
  }
)

let myMap := [myList mapBy : "name"]
display myMap
```

outputs:

```
map: @[
  "Arnold" :>
    struct: @{
      age :>
        integer: 18
      height :>
        integer: 180
      name :>
        string: "Arnold"
    }
  "Bob" :>
    struct: @{
      age :>
        integer: 22
```

```

        height :>
            integer: 170
        name :>
            string: "Bob"
    }
    "John" :>
        struct: @{
            age :>
                integer: 29
            height :>
                integer: 175
            name :>
                string: "John"
        }
]

```

## 1.8 The map data type

The map data type allows to store an association of key and value. map members are accessed by using `[<key>]` where `<key>` is a string. If `A` is a map, `A["John"]` refers to element of `A` having key "John".

A literal map is defined as follow:

```
@[ "age" : 29, "height" : 175, "name" : "John" ]
```

An empty map can be initialized using the `emptymap` constant.

 The `emptymap` constant is deprecated. Use a literal empty map, `@[]`, instead.

### 1.8.1 map operators

The map data type supports the following operators:

Operator	Expression type	Meaning
<code>!=</code>	<code>bool ← map != map</code>	Not equal
<code>==</code>	<code>bool ← map == map</code>	Equal

Two maps are equal if:

- they have the same number of items
- they have the same item keys
- they have the same item values

## 2 GTL instructions

### 2.1 The *let* instruction

Data assignment instruction. The general form is:

```
let var := expression
```

As in the C language, GTL has assignment operator:

Assign.	int	float	string	bool	struct	list	map	uncons
+=	+	+	concat	NA	NA	add	NA	NA
-=	-	-	NA	NA	NA	NA	NA	NA
*=	*	*	NA	NA	NA	NA	NA	NA
/=	/	/	NA	NA	NA	NA	NA	NA
mod=	mod	NA	NA	NA	NA	NA	NA	NA
<<=	<<	NA	NA	NA	NA	NA	NA	NA
>>=	>>	NA	NA	NA	NA	NA	NA	NA
&=	bitwise &	NA	NA	logical &	NA	NA	NA	NA
=	bitwise	NA	NA	logical	NA	NA	NA	NA
^=	bitwise ^	NA	NA	logical ^	NA	NA	NA	NA

The scope of a variable depends on the location where the variable is assigned the first time. For instance, in the following code:

```
let a := 1
foreach task in TASKS do
  let b := 2
  let a += 1
end foreach
println a
println b
```

Because `a` is assigned outside the `foreach` loop, it is both accessible within the `foreach` loop and accessible after the `foreach` loop. So it contains the value number of items in `TASKS` + 1 after the `foreach`. Because `b` is assigned inside the `foreach` loop, it does not exist after the loop anymore and `println b` will trigger an error.

### 2.2 The *if* instruction

Conditional execution. The forms are:

```
if expression then ... end if
if expression then ... else ... end if
if expression then ... elsif expression then ... end if
if expression then ... elsif expression then ... else ... end if
```

The *expression* must be boolean. In the following example, the blue text (within the `%`) is produced only if the `USECOM` boolean variable is true:

```

if USECOM then %
#include "tpl_com.h" %
end if

```

## 2.3 The *foreach* instruction

This instruction iterates on the elements of a collection, a list or a map. The simplest form is the following one:

```
foreach var in expression do ... end foreach
```

Here *var* takes the value of each of the elements of the collection. If the collection is a list, the elements are iterated in the order of the list. If the collection is a map, the element are iterated in the alphanumerical order of the keys. In both cases, a variable named *INDEX* which contains the current iteration number is available inside the loop. *INDEX* ranges from 0 to the number of elements in the list minus 1. If the collection is a map, a second variable, *KEY*, which contains the key associated to the value of the current item, is available.

In the following example, for each element in the *ALARMS* list, the text between the *do* and the *end foreach* is produced with the *NAME* attribute of the current element of the *ALARMS* list inserted at the specified location.

```

foreach alr in ALARMS do
%
/* Alarm % !alr::NAME % identifier */
#define % !alr::NAME %_id % !INDEX %
CONST(AlarmType, AUTOMATIC) % !alr::NAME % = % !NAME %_id;
%
end foreach

```

A more general form of the *foreach* instruction is:

```

foreach key,var (index) in expression
  before ...
  do ...
  between ...
  after ...
end foreach

```

*key* may be used only when iterating on a map and allows to give a custom name to the default *KEY* variable. (*index*) may be used both for a list or a map and allows to give a custom name to the default *INDEX* variable.

If the collection is not empty, the **before** section is executed once before the first execution of the **do** section. If the collection contains at least two elements, the **between** section is executed between the execution of the **do** section. If the list is not empty, the **after** section is executed once after the last execution of the **do** section.

The following example illustrates the general form. Here a table of pointers to alarm descriptors is generated:



```

#
# Initialize ALARMS with a list of 2 structs with a NAME field.
#
let ALARMS := @( @{ NAME: "alr1"}, @{ NAME: "alr1"} )

foreach alr in ALARMS
  before %
  tpl_time_obj *tpl_alarm_table[ALARM_COUNT] = {
  %
  do %  &% !alr::NAME %_alarm_desc%
  between %,
  %
  after %
};
%
end foreach

```

It produces the following output:

```

tpl_time_obj *tpl_alarm_table[ALARM_COUNT] = {
  &alr1_alarm_desc,
  &alr1_alarm_desc
};

```

## 2.4 The *for* instruction

The **for** instruction iterates along a literal list of elements.

```

for var in expression, ... , expression do
  ...
end for

```

At each iteration, *var* gets the value of the current *expression*. As in the **foreach** instruction, *INDEX* is generated and ranges from 0 to the number of elements in the list minus 1.

 The *for* instruction is deprecated. Use **foreach** with a literal list instead.

## 2.5 The *loop* instruction

The **loop** instruction iterate over a range of integers. Its simplest form is:

```

loop var from expression_start to expression_end do
  ...
end loop

```

Both *expression\_start* and *expression\_end* must be integer expression. By default *var* is incremented by one from *expression\_start*, inclusive, to *expression\_end*, inclusive.

Like in the `foreach` instruction, `before`, `between` and `after` sections may be used. Moreover, `down` may be used to decrement `var` by one. `step` allows to increment or decrement by `increment`.

```
loop var from expression <down> to expression <step> <increment>
  before ...
  do ...
  between ...
  after ...
end loop
```

For instance, in the following loop, `var` goes from 0 to 10 with an increment of 2:

```
loop a from 0 to 10 step 2 do
  display a
end loop
```

and produces the following output:

```
integer: 0
integer: 2
integer: 4
integer: 6
integer: 8
integer: 10
```